// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)

pragma solidity ^0.8.20;

import {Context} from "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * The initial owner is set to the address provided by the deployer. This can
 * later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    /**
     * @dev The caller account is not authorized to perform an operation.
     */
    error OwnableUnauthorizedAccount(address account);

    /**
     * @dev The owner is not a valid owner account. (eg. `address(0)`)
     */
    error OwnableInvalidOwner(address owner);

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
     */
    constructor(address initialOwner) {
        if (initialOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(initialOwner);
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        if (owner() != _msgSender()) {
            revert OwnableUnauthorizedAccount(_msgSender());
        }
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        if (newOwner == address(0)) {
            revert OwnableInvalidOwner(address(0));
        }
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.20;

import {IERC20} from "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.20;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the value of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the value of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves a `value` amount of tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 value) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets a `value` amount of tokens as the allowance of `spender` over the
     * caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 value) external returns (bool);

    /**
     * @dev Moves a `value` amount of tokens from `from` to `to` using the
     * allowance mechanism. `value` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 value) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.1) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }

    function _contextSuffixLength() internal view virtual returns (uint256) {
        return 0;
    }
}

// File: @openzeppelin/contracts/utils/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/Pausable.sol)

pragma solidity ^0.8.20;

import {Context} from "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    bool private _paused;

    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    /**
     * @dev The operation failed because the contract is paused.
     */
    error EnforcedPause();

    /**
     * @dev The operation failed because the contract is not paused.
     */
    error ExpectedPause();

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        if (paused()) {
            revert EnforcedPause();
        }
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        if (!paused()) {
            revert ExpectedPause();
        }
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/utils/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v5.0.0) (utils/ReentrancyGuard.sol)

pragma solidity ^0.8.20;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant NOT_ENTERED = 1;
    uint256 private constant ENTERED = 2;

    uint256 private _status;

    /**
     * @dev Unauthorized reentrant call.
     */
    error ReentrancyGuardReentrantCall();

    constructor() {
        _status = NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be NOT_ENTERED
        if (_status == ENTERED) {
            revert ReentrancyGuardReentrantCall();
        }

        // Any calls to nonReentrant after this point will fail
        _status = ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == ENTERED;
    }
}

// File: contracts/ERC20AirSwapEscrow.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity ^0.8.23;

// Uncomment this line to use console.log
// import "hardhat/console.sol";

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol";
// import "@openzeppelin/contracts/utils/Arrays.sol";
// import "@openzeppelin/contracts/utils/math/Math.sol";
// import "@openzeppelin/contracts/utils/math/SafeCast.sol";
// import "@openzeppelin/contracts/utils/math/SignedMath.sol";
import "@openzeppelin/contracts/utils/Pausable.sol";
import "@openzeppelin/contracts/utils/ReentrancyGuard.sol";
// import "@openzeppelin/contracts/utils/Strings.sol";
// import "@openzeppelin/contracts/utils/structs/EnumerableSet.sol";

contract ERC20AirSwapEscrow is Ownable, Pausable, ReentrancyGuard
{
    // using Math for uint256;
    // using SignedMath for int256;
    // using Strings for string;

    enum OfferStatus { 
        available,
        taken,
        retracted
    }

    struct SellOffer {
        address seller;
        uint256 amount;
        uint256 priceAmount;
        address priceERC20;
        uint256 postedTimestamp;
        OfferStatus status;
        address buyer;
        uint256 closedTimestamp;
    }

    struct BuyOffer {
        address buyer;
        uint256 amount;
        uint256 priceAmount;
        address priceERC20;
        uint256 postedTimestamp;
        OfferStatus status;
        address seller;
        uint256 closedTimestamp;
    }

    address private _swappedERC20;
    SellOffer[] private _sellOffers;
    BuyOffer[] private _buyOffers;

    event SellOfferPosted(uint256 offerIndex, SellOffer offer);
    event SellOfferRetracted(uint256 offerIndex, SellOffer offer);
    event SellOfferTaken(uint256 offerIndex, SellOffer offer);
    event BuyOfferPosted(uint256 offerIndex, BuyOffer offer);
    event BuyOfferRetracted(uint256 offerIndex, BuyOffer offer);
    event BuyOfferTaken(uint256 offerIndex, BuyOffer offer);

    constructor(address swappedERC20_) Ownable(msg.sender) Pausable() ReentrancyGuard() {
        require(swappedERC20_ != address(0), "Swapped ERC20 token address cannot be null!");
        _swappedERC20 = swappedERC20_;
    }



    // private token functions

    function _receiveToken(address from, address token, uint256 amount) private {
        if (token == address(0)) {
            require(msg.value >= amount, "Not enough received native token!");
        }
        else {
            require(IERC20(token).transferFrom(from, address(this), amount), "Not enough received ERC20 token!");
        }
    }

    function _sendToken(address to, address token, uint256 amount) private {
        if (token == address(0)) {
            require(payable(to).send(amount), "Not enough native token to send!");
        }
        else {
            require(IERC20(token).transfer(to, amount), "Not enough ERC20 token to send!");
        }
    }



    // private escrow functions

    function _createSellOffer(address seller, uint256 amount, uint256 priceAmount, address priceERC20) private returns (uint256, SellOffer storage) {
        _sellOffers.push(SellOffer(seller, amount, priceAmount, priceERC20, block.timestamp, OfferStatus.available, address(0), 0));
        return (_sellOffers.length - 1, _sellOffers[_sellOffers.length - 1]);
    }

    function _getSellOffer(uint256 offerIndex) private view returns (SellOffer storage) {
        require(offerIndex < _sellOffers.length, "Sell offer index out out bounds!");
        return _sellOffers[offerIndex];
    }

    function _filterSellOffers(bool bySeller, address ofSeller, bool byStatus, OfferStatus ofStatus, bool byBuyer, address ofBuyer) private view returns (SellOffer[] memory) {
        uint256 specificSellOffersCount = 0;
        uint256[] memory specificSellOfferIds = new uint256[](_sellOffers.length);
        for (uint256 i = 0; i < _sellOffers.length; i++) {
            if ((!bySeller || _sellOffers[i].seller == ofSeller) && (!byStatus || _sellOffers[i].status == ofStatus) && (!byBuyer || _sellOffers[i].buyer == ofBuyer)) {
                specificSellOfferIds[i] = i;
                specificSellOffersCount++;
            }
        }
        SellOffer[] memory specificSellOffers = new SellOffer[](specificSellOffersCount);
        for (uint256 i = 0; i < specificSellOffersCount; i++) {
            specificSellOffers[i] = _sellOffers[specificSellOfferIds[i]];
        }
        return specificSellOffers;
    }

    function _closeSellOffer(address closer, SellOffer storage offer, OfferStatus closingStatus) private {
        require(offer.status == OfferStatus.available, "Sell offer not available anymore!");
        if (closingStatus == OfferStatus.retracted) {
            require(offer.seller == closer, "Cannot retract sell offer of a different seller!");
        }
        offer.status = closingStatus;
        offer.closedTimestamp = block.timestamp;
        if (closingStatus == OfferStatus.taken) {
            offer.buyer = closer;
        }
    }

    function _createBuyOffer(address buyer, uint256 amount, uint256 priceAmount, address priceERC20) private returns (uint256, BuyOffer storage) {
        _buyOffers.push(BuyOffer(buyer, amount, priceAmount, priceERC20, block.timestamp, OfferStatus.available, address(0), 0));
        return (_buyOffers.length - 1, _buyOffers[_buyOffers.length - 1]);
    }

    function _getBuyOffer(uint256 offerIndex) private view returns (BuyOffer storage) {
        require(offerIndex < _buyOffers.length, "Buy offer index out out bounds!");
        return _buyOffers[offerIndex];
    }

    function _filterBuyOffers(bool byBuyer, address ofBuyer, bool byStatus, OfferStatus ofStatus, bool bySeller, address ofSeller) private view returns (BuyOffer[] memory) {
        uint256 specificBuyOffersCount = 0;
        uint256[] memory specificBuyOfferIds = new uint256[](_buyOffers.length);
        for (uint256 i = 0; i < _buyOffers.length; i++) {
            if ((!byBuyer || _buyOffers[i].buyer == ofBuyer) && (!byStatus || _buyOffers[i].status == ofStatus) && (!bySeller || _buyOffers[i].seller == ofSeller)) {
                specificBuyOfferIds[i] = i;
                specificBuyOffersCount++;
            }
        }
        BuyOffer[] memory specificBuyOffers = new BuyOffer[](specificBuyOffersCount);
        for (uint256 i = 0; i < specificBuyOffersCount; i++) {
            specificBuyOffers[i] = _buyOffers[specificBuyOfferIds[i]];
        }
        return specificBuyOffers;
    }

    function _closeBuyOffer(address closer, BuyOffer storage offer, OfferStatus closingStatus) private {
        require(offer.status == OfferStatus.available, "Buy offer not available anymore!");
        if (closingStatus == OfferStatus.retracted) {
            require(offer.buyer == closer, "Cannot retract buy offer of a different buyer!");
        }
        offer.status = closingStatus;
        offer.closedTimestamp = block.timestamp;
        if (closingStatus == OfferStatus.taken) {
            offer.seller = closer;
        }
    }



    // public pause/unpause functions

    function pause() public onlyOwner {
        _pause();
    }

    function unpause() public onlyOwner {
        _unpause();
    }



    // public escrow read functions

    function swappedERC20() public view returns (address) {
        return _swappedERC20;
    }

    function swappedERC20Name() public view returns (string memory) {
        return IERC20Metadata(_swappedERC20).name();
    }

    function swappedERC20Symbol() public view returns (string memory) {
        return IERC20Metadata(_swappedERC20).symbol();
    }

    function swappedERC20Decimals() public view returns (uint8) {
        return IERC20Metadata(_swappedERC20).decimals();
    }

    function sellOffer(uint256 offerIndex) public view returns (SellOffer memory) {
        return _getSellOffer(offerIndex);
    }

    function sellOfferCount() public view returns (uint256) {
        return _sellOffers.length;
    }

    function sellOffers() public view returns (SellOffer[] memory) {
        return _sellOffers;
    }

    function filteredSellOffers(bool bySeller, address ofSeller, bool byStatus, OfferStatus ofStatus, bool byBuyer, address ofBuyer) public view returns (SellOffer[] memory) {
        return _filterSellOffers(bySeller, ofSeller, byStatus, ofStatus, byBuyer, ofBuyer);
    }

    function buyOffer(uint256 offerIndex) public view returns (BuyOffer memory) {
        return _getBuyOffer(offerIndex);
    }

    function buyOfferCount() public view returns (uint256) {
        return _buyOffers.length;
    }

    function buyOffers() public view returns (BuyOffer[] memory) {
        return _buyOffers;
    }

    function filteredBuyOffers(bool byBuyer, address ofBuyer, bool byStatus, OfferStatus ofStatus, bool bySeller, address ofSeller) public view returns (BuyOffer[] memory) {
        return _filterBuyOffers(byBuyer, ofBuyer, byStatus, ofStatus, bySeller, ofSeller);
    }



    // public escrow write functions

    function postSellOffer(uint256 amount, uint256 priceAmount, address priceERC20) public whenNotPaused nonReentrant returns (SellOffer memory) {
        _receiveToken(msg.sender, _swappedERC20, amount);
        (uint256 offerIndex, SellOffer storage offer) = _createSellOffer(msg.sender, amount, priceAmount, priceERC20);
        emit SellOfferPosted(offerIndex, offer);
        return offer;
    }

    function retractSellOffer(uint256 offerIndex) public whenNotPaused nonReentrant returns (SellOffer memory) {
        SellOffer storage offer = _getSellOffer(offerIndex);
        _closeSellOffer(msg.sender, offer, OfferStatus.retracted);
        _sendToken(offer.seller, _swappedERC20, offer.amount);
        emit SellOfferRetracted(offerIndex, offer);
        return offer;
    }

    function takeSellOffer(uint256 offerIndex) public payable whenNotPaused nonReentrant returns (SellOffer memory) {
        SellOffer storage offer = _getSellOffer(offerIndex);
        _receiveToken(msg.sender, offer.priceERC20, offer.priceAmount);
        _closeSellOffer(msg.sender, offer, OfferStatus.taken);
        _sendToken(offer.seller, offer.priceERC20, offer.priceAmount);
        _sendToken(offer.buyer, _swappedERC20, offer.amount);
        emit SellOfferTaken(offerIndex, offer);
        return offer;
    }

    function postBuyOffer(uint256 amount, uint256 priceAmount, address priceERC20) public payable whenNotPaused nonReentrant returns (BuyOffer memory) {
        _receiveToken(msg.sender, priceERC20, priceAmount);
        (uint256 offerIndex, BuyOffer storage offer) = _createBuyOffer(msg.sender, amount, priceAmount, priceERC20);
        emit BuyOfferPosted(offerIndex, offer);
        return offer;
    }

    function retractBuyOffer(uint256 offerIndex) public whenNotPaused nonReentrant returns (BuyOffer memory) {
        BuyOffer storage offer = _getBuyOffer(offerIndex);
        _closeBuyOffer(msg.sender, offer, OfferStatus.retracted);
        _sendToken(offer.buyer, offer.priceERC20, offer.priceAmount);
        return offer;
    }

    function takeBuyOffer(uint256 offerIndex) public whenNotPaused nonReentrant returns (BuyOffer memory) {
        BuyOffer storage offer = _getBuyOffer(offerIndex);
        _receiveToken(msg.sender, _swappedERC20, offer.amount);
        _closeBuyOffer(msg.sender, offer, OfferStatus.taken);
        _sendToken(offer.buyer, _swappedERC20, offer.amount);
        _sendToken(offer.seller, offer.priceERC20, offer.priceAmount);
        return offer;
    }



    // external funds functions

    function balanceOf(address token) external view returns (uint256) {
        if (token == address(0)) {
            return address(this).balance;
        }
        return IERC20(token).balanceOf(address(this));
    }

    function depositToken(address token, uint256 amount) external payable onlyOwner {
        _receiveToken(msg.sender, token, amount);
    }

    function withdrawToken(address to, address token, uint256 amount) external onlyOwner {
        _sendToken(to, token, amount);
    }
}